

NAG Library Routine Document

F08BTF (ZGEQP3)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

1 Purpose

F08BTF (ZGEQP3) computes the QR factorization, with column pivoting, of a complex m by n matrix.

2 Specification

```
SUBROUTINE F08BTF (M, N, A, LDA, JPVT, TAU, WORK, LWORK, RWORK, INFO)
INTEGER          M, N, LDA, JPVT(*), LWORK, INFO
REAL (KIND=nag_wp) RWORK(*)
COMPLEX (KIND=nag_wp) A(LDA,*), TAU(*), WORK(max(1,LWORK))
```

The routine may be called by its LAPACK name *zgeqp3*.

3 Description

F08BTF (ZGEQP3) forms the QR factorization, with column pivoting, of an arbitrary rectangular complex m by n matrix.

If $m \geq n$, the factorization is given by:

$$AP = Q \begin{pmatrix} R \\ 0 \end{pmatrix},$$

where R is an n by n upper triangular matrix (with real diagonal elements), Q is an m by m unitary matrix and P is an n by n permutation matrix. It is sometimes more convenient to write the factorization as

$$AP = (Q_1 \quad Q_2) \begin{pmatrix} R \\ 0 \end{pmatrix},$$

which reduces to

$$AP = Q_1 R,$$

where Q_1 consists of the first n columns of Q , and Q_2 the remaining $m - n$ columns.

If $m < n$, R is trapezoidal, and the factorization can be written

$$AP = Q (R_1 \quad R_2),$$

where R_1 is upper triangular and R_2 is rectangular.

The matrix Q is not formed explicitly but is represented as a product of $\min(m, n)$ elementary reflectors (see the F08 Chapter Introduction for details). Routines are provided to work with Q in this representation (see Section 9).

Note also that for any $k < n$, the information returned in the first k columns of the array A represents a QR factorization of the first k columns of the permuted matrix AP .

The routine allows specified columns of A to be moved to the leading columns of AP at the start of the factorization and fixed there. The remaining columns are free to be interchanged so that at the i th stage the pivot column is chosen to be the column which maximizes the 2-norm of elements i to m over columns i to n .

4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia <http://www.netlib.org/lapack/lug>

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

5 Arguments

- 1: M – INTEGER *Input*
On entry: m , the number of rows of the matrix A .
Constraint: $M \geq 0$.
- 2: N – INTEGER *Input*
On entry: n , the number of columns of the matrix A .
Constraint: $N \geq 0$.
- 3: A(LDA,*) – COMPLEX (KIND=nag_wp) array *Input/Output*
Note: the second dimension of the array A must be at least $\max(1, N)$.
On entry: the m by n matrix A .
On exit: if $m \geq n$, the elements below the diagonal are overwritten by details of the unitary matrix Q and the upper triangle is overwritten by the corresponding elements of the n by n upper triangular matrix R .
 If $m < n$, the strictly lower triangular part is overwritten by details of the unitary matrix Q and the remaining elements are overwritten by the corresponding elements of the m by n upper trapezoidal matrix R .
 The diagonal elements of R are real.
- 4: LDA – INTEGER *Input*
On entry: the first dimension of the array A as declared in the (sub)program from which F08BTF (ZGEQP3) is called.
Constraint: $LDA \geq \max(1, M)$.
- 5: JPVT(*) – INTEGER array *Input/Output*
Note: the dimension of the array JPVT must be at least $\max(1, N)$.
On entry: if $JPVT(j) \neq 0$, then the j th column of A is moved to the beginning of AP before the decomposition is computed and is fixed in place during the computation. Otherwise, the j th column of A is a free column (i.e., one which may be interchanged during the computation with any other free column).
On exit: details of the permutation matrix P . More precisely, if $JPVT(j) = k$, then the k th column of A is moved to become the j th column of AP ; in other words, the columns of AP are the columns of A in the order $JPVT(1), JPVT(2), \dots, JPVT(n)$.
- 6: TAU(*) – COMPLEX (KIND=nag_wp) array *Output*
Note: the dimension of the array TAU must be at least $\max(1, \min(M, N))$.
On exit: further details of the unitary matrix Q .

- 7: WORK(max(1,LWORK)) – COMPLEX (KIND=nag_wp) array Workspace
On exit: if INFO = 0, the real part of WORK(1) contains the minimum value of LWORK required for optimal performance.
- 8: LWORK – INTEGER Input
On entry: the dimension of the array WORK as declared in the (sub)program from which F08BTF (ZGEP3) is called.
 If LWORK = -1, a workspace query is assumed; the routine only calculates the optimal size of the WORK array, returns this value as the first entry of the WORK array, and no error message related to LWORK is issued.
Suggested value: for optimal performance, $LWORK \geq (N + 1) \times nb$, where nb is the optimal **block size**.
Constraint: $LWORK \geq N + 1$ or $LWORK = -1$.
- 9: RWORK(*) – REAL (KIND=nag_wp) array Workspace
Note: the dimension of the array RWORK must be at least $\max(1, 2 \times N)$.
- 10: INFO – INTEGER Output
On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

INFO < 0

If INFO = - i , argument i had an illegal value. An explanatory message is output, and execution of the program is terminated.

7 Accuracy

The computed factorization is the exact factorization of a nearby matrix $(A + E)$, where

$$\|E\|_2 = O(\epsilon)\|A\|_2,$$

and ϵ is the *machine precision*.

8 Parallelism and Performance

F08BTF (ZGEP3) is threaded by NAG for parallel execution in multithreaded implementations of the NAG Library.

F08BTF (ZGEP3) makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

9 Further Comments

The total number of real floating-point operations is approximately $\frac{8}{3}n^2(3m - n)$ if $m \geq n$ or $\frac{8}{3}m^2(3n - m)$ if $m < n$.

To form the unitary matrix Q F08BTF (ZGEP3) may be followed by a call to F08ATF (ZUNGQR):

```
CALL ZUNGQR(M,M,MIN(M,N),A,LDA,TAU,WORK,LWORK,INFO)
```

but note that the second dimension of the array A must be at least M , which may be larger than was required by F08BTF (ZGEQP3).

When $m \geq n$, it is often only the first n columns of Q that are required, and they may be formed by the call:

```
CALL ZUNGQR(M,N,N,A,LDA,TAU,WORK,LWORK,INFO)
```

To apply Q to an arbitrary complex rectangular matrix C , F08BTF (ZGEQP3) may be followed by a call to F08AUF (ZUNMQR). For example,

```
CALL ZUNMQR('Left','Conjugate Transpose',M,P,MIN(M,N),A,LDA,TAU, &
           C,LDC,WORK,LWORK,INFO)
```

forms $C = Q^H C$, where C is m by p .

To compute a QR factorization without column pivoting, use F08ASF (ZGEQRF).

The real analogue of this routine is F08BFF (DGEQP3).

10 Example

This example solves the linear least squares problems

$$\min_x \|b_j - Ax_j\|_2, \quad j = 1, 2$$

for the basic solutions x_1 and x_2 , where

$$A = \begin{pmatrix} 0.47 - 0.34i & -0.40 + 0.54i & 0.60 + 0.01i & 0.80 - 1.02i \\ -0.32 - 0.23i & -0.05 + 0.20i & -0.26 - 0.44i & -0.43 + 0.17i \\ 0.35 - 0.60i & -0.52 - 0.34i & 0.87 - 0.11i & -0.34 - 0.09i \\ 0.89 + 0.71i & -0.45 - 0.45i & -0.02 - 0.57i & 1.14 - 0.78i \\ -0.19 + 0.06i & 0.11 - 0.85i & 1.44 + 0.80i & 0.07 + 1.14i \end{pmatrix}$$

and

$$B = \begin{pmatrix} -1.08 - 2.59i & 2.22 + 2.35i \\ -2.61 - 1.49i & 1.62 - 1.48i \\ 3.13 - 3.61i & 1.65 + 3.43i \\ 7.33 - 8.01i & -0.98 + 3.08i \\ 9.12 + 7.63i & -2.84 + 2.78i \end{pmatrix}.$$

and b_j is the j th column of the matrix B . The solution is obtained by first obtaining a QR factorization with column pivoting of the matrix A . A tolerance of 0.01 is used to estimate the rank of A from the upper triangular factor, R .

Note that the block size (NB) of 64 assumed in this example is not realistic for such a small problem, but should be suitable for large problems.

10.1 Program Text

```
Program f08btfe
!      F08BTF Example Program Text
!
!      Mark 26 Release. NAG Copyright 2016.
!
!      .. Use Statements ..
Use nag_library, Only: dznrm2, nag_wp, x04dbf, zgeqp3, ztrsm, zunmqr
!      .. Implicit None Statement ..
Implicit None
!      .. Parameters ..
Complex (Kind=nag_wp), Parameter :: one = (1.0E0_nag_wp,0.0E0_nag_wp)
Complex (Kind=nag_wp), Parameter :: zero = (0.0E0_nag_wp,0.0E0_nag_wp)
Integer, Parameter                :: incl = 1, nb = 64, nin = 5, nout = 6
!      .. Local Scalars ..
Real (Kind=nag_wp)                :: tol
Integer                            :: i, ifail, info, j, k, lda, ldb,      &
```

```

                                lwork, m, n, nrhs
!   .. Local Arrays ..
Complex (Kind=nag_wp), Allocatable :: a(:,,:), b(:,,:), tau(:), work(:)
Real (Kind=nag_wp), Allocatable  :: rnorm(:), rwork(:)
Integer, Allocatable              :: jpvt(:)
Character (1)                     :: clabs(1), rlabs(1)
!   .. Intrinsic Procedures ..
Intrinsic                          :: abs
!   .. Executable Statements ..
Write (nout,*) 'F08BTF Example Program Results'
Write (nout,*)
!   Skip heading in data file
Read (nin,*)
Read (nin,*) m, n, nrhs
lda = m
ldb = m
lwork = (n+1)*nb
Allocate (a(lda,n),b(ldb,nrhs),tau(n),work(lwork),rnorm(nrhs),      &
         rwork(2*n),jpvt(n))

!   Read A and B from data file

Read (nin,*)(a(i,1:n),i=1,m)
Read (nin,*)(b(i,1:nrhs),i=1,m)

!   Initialize JPVT to be zero so that all columns are free

jpvt(1:n) = 0

!   Compute the QR factorization of A

!   The NAG name equivalent of zgeqp3 is f08btf
Call zgeqp3(m,n,a,lda,jpvt,tau,work,lwork,rwork,info)

!   Compute C = (C1) = (Q**H)*B, storing the result in B
!               (C2)

!   The NAG name equivalent of zunmqr is f08auf
Call zunmqr('Left','Conjugate Transpose',m,nrhs,n,a,lda,tau,b,ldb,work, &
         lwork,info)

!   Choose TOL to reflect the relative accuracy of the input data

tol = 0.01_nag_wp

!   Determine and print the rank, K, of R relative to TOL

loop: Do k = 1, n
    If (abs(a(k,k))<=tol*abs(a(1,1))) Then
        Exit loop
    End If
End Do loop
k = k - 1

Write (nout,*) 'Tolerance used to estimate the rank of A'
Write (nout,99999) tol
Write (nout,*) 'Estimated rank of A'
Write (nout,99998) k
Write (nout,*)
Flush (nout)

!   Compute least squares solutions by back-substitution in
!   R(1:K,1:K)*Y = C1, storing the result in B

!   The NAG name equivalent of ztrsm is f06zjf
Call ztrsm('Left','Upper','No transpose','Non-Unit',k,nrhs,one,a,lda,b, &
         ldb)

!   Compute estimates of the square roots of the residual sums of
!   squares (2-norm of each of the columns of C2)

```

```

! The NAG name equivalent of dznrm2 is f06jjf
Do j = 1, nrhs
  rnorm(j) = dznrm2(m-k,b(k+1,j),incl)
End Do

! Set the remaining elements of the solutions to zero (to give
! the basic solutions)

b(k+1:n,1:nrhs) = zero

! Permute the least squares solutions stored in B to give X = P*Y

Do j = 1, nrhs
  work(jpvt(1:n)) = b(1:n,j)
  b(1:n,j) = work(1:n)
End Do

! Print least squares solutions

! ifail: behaviour on error exit
!       =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
ifail = 0
Call x04dbf('General',' ',n,nrhs,b,ldb,'Bracketed','F7.4',      &
  'Least squares solution(s)','Integer',rlabs,'Integer',clabs,80,0, &
  ifail)

! Print the square roots of the residual sums of squares

Write (nout,*)
Write (nout,*) 'Square root(s) of the residual sum(s) of squares'
Write (nout,99999) rnorm(1:nrhs)

99999 Format (3X,1P,7E11.2)
99998 Format (1X,I8)
End Program f08btfe

```

10.2 Program Data

F08BTF Example Program Data

```

5           4           2           :Values of M, N and NRHS

( 0.47,-0.34) (-0.40, 0.54) ( 0.60, 0.01) ( 0.80,-1.02)
(-0.32,-0.23) (-0.05, 0.20) (-0.26,-0.44) (-0.43, 0.17)
( 0.35,-0.60) (-0.52,-0.34) ( 0.87,-0.11) (-0.34,-0.09)
( 0.89, 0.71) (-0.45,-0.45) (-0.02,-0.57) ( 1.14,-0.78)
(-0.19, 0.06) ( 0.11,-0.85) ( 1.44, 0.80) ( 0.07, 1.14) :End of matrix A

(-1.08,-2.59) ( 2.22, 2.35)
(-2.61,-1.49) ( 1.62,-1.48)
( 3.13,-3.61) ( 1.65, 3.43)
( 7.33,-8.01) (-0.98, 3.08)
( 9.12, 7.63) (-2.84, 2.78) :End of matrix B

```

10.3 Program Results

F08BTF Example Program Results

```

Tolerance used to estimate the rank of A
1.00E-02
Estimated rank of A
3

Least squares solution(s)
1           1           2
1 ( 0.0000, 0.0000) ( 0.0000, 0.0000)
2 ( 2.7020, 8.0911) (-2.2682,-2.9884)

```

```
3 ( 2.8888, 2.5012) ( 0.9779, 1.3565)
4 ( 2.7100, 0.4791) (-1.3734, 0.2212)
```

```
Square root(s) of the residual sum(s) of squares
2.51E-01 8.10E-02
```
